PROPERTIES OF DELÉHAM'S DELTA TRANSFORMATION: OEIS A084938

RICHARD J. MATHAR

Abstract. Deléham's Delta-Transformation generates a number triangle from two number sequences by providing a generating function closely related to the bivariate infinite continued fraction of the two number sequences. We mainly summarize recovery of the bivariate generating function of the triangle from the univariate generating functions of the two sequences.

1. Definitions and Notation

In A084938[13, A084938] Deléham defines a set of lower infinite number triangles $T_{n,k}$ with row indices $n \ge 0$ and column indices $0 \le k \le n$ by a pair of generating number sequences (r_i, s_i), $i \ge 0$, as follows: define an auxiliary array of polynomials $P_{n,m}(x,y)$ recursively as

$$(1)\ P_{0,m}(x,y) = 1, \quad m \ge 0$$

$$(2)\ P_{n,-1}(x,y) = 0, \quad n \ge 0$$

$$(3)\ P_{n,m}(x,y) = P_{n,m-1}(x,y) + (r_m x + s_m y) P_{n-1,m+1}(x,y), \quad n \ge 1, m \ge 0.$$

Then

$$T_{n,k} = [x^{n-k} y^k] P_{n,0}(x,y), \quad 0 \le k \le n, \quad n \ge 0. \tag{4}$$

The polynomials $P_{n,m}(x,y)$ in row n are homogeneous polynomials of order n. If the first N of the elements of r_i and s_i are known, the polynomials $P_{n,m}$ are known in the upper left triangle $n \le N$, $0 \le m \le N-n$, and this suffices to generate the elements of the triangle $T_{n,k}$ up row $n \le N$.

There are two views for a Maple implementation. The first one is to use r and s as arguments and ask for a single element $T_{n,k}$ of the triangle:

```
Delta := proc(r::list,s::list,n,k)
        option remember;
        local P ,nloc,kloc,q,N,x,y;
        N := min(nops(r),nops(s)) ;
        P := Array(0..N,0..N) ;
        for kloc from 0 to N do
                P[0,kloc] := 1 ;
        end do:
        for nloc from 1 to min(N,n) do
                for kloc from 0 to N-nloc-1 do
                        q := x*op(1+kloc,r)+y*op(1+kloc,s) ;
                        if kloc > 0 then
```

Date: August 23, 2015.
2010 *Mathematics Subject Classification.* Primary 15B36; Secondary 26C15, A41A21.
Key words and phrases. Sequences, Lower Triangular Array, Delta Transformation.

```
                                P[nloc,kloc] := expand(P[nloc,kloc-1]+q*P[nloc-1,kloc+1]) ;
                        else
                                P[nloc,kloc] := expand(q*P[nloc-1,kloc+1]) ;
                        end if;
                end do:
        end do:
        coeff(coeff(P[n,0],x,n-k),y,k) ;
end proc:
```

The other one is to print all elements up to the point where the smaller length of the two r and s lists exhausts the output:

```
DELTA := proc(r::list,s::list)
        local P ,nloc,kloc,q,N,x,y;
        N := min(nops(r),nops(s)) ;
        P := Array(0..N,0..N) ;
        for kloc from 0 to N do
                P[0,kloc] := 1 ;
        end do:
        printf("1,\n") ;
        for nloc from 1 to N do
                for kloc from 0 to N-nloc-1 do
                        q := x*op(1+kloc,r)+y*op(1+kloc,s) ;
                        if kloc > 0 then
                                P[nloc,kloc] := expand(P[nloc,kloc-1]+q*P[nloc-1,kloc+1]) ;
                        else
                                P[nloc,kloc] := expand(q*P[nloc-1,kloc+1]) ;
                        end if;
                        if kloc = 0 then
                                for k from 0 to nloc do
                                        printf("%d,",coeff(coeff(P[nloc,0],x,nloc-k),y,k) );
                                end do;
                        end if;
                end do:
                printf("\n") ;
        end do:
        return ;
end proc:
```

2. Symmetries

Theorem 1. *Swapping the role of the r and s with $r_i \leftrightarrow s_i$ reverses the order of elements in each row of T with $T_{n,k} \leftrightarrow T_{n,n-k}$.*

Theorem 2. *If all $r_i = 0$, all T are zero in the subdiagonals: $T_{n,k} = 0$ for $0 \le k < n$.*

Theorem 3. *If all $s_i = 0$, all T are zero outside the first column: $T_{n,k} = 0$ for $k > 0$.*

The two degenerate cases of populating only the first column or the diagonal are not interesting because they are better represented by univariate plain sequences. Examples are A127647, A127648, A131427, A134309, A198954, or the triangular interpretation of A010054.

3. Bivariate Generating Function

3.1. **General Form.** The bivariate generating function g of the triangle in the variables x and y is defined as

Definition 1.

$$g(z,t) \equiv \sum_{n\geq 0,k\geq 0} T_{n,k} z^n t^k. \tag{5}$$

Theorem 4. *The bivariate generating function of (4) is related to the generating series by the infinite continued fraction*

$$g(z,t) = \cfrac{1}{1-\cfrac{r_0z+s_0zt}{1-\cfrac{r_1z+s_1zt}{1-\cfrac{r_2z+s_2zt}{1-\cdots}}}}. \tag{6}$$

3.2. **Finite Left and Right sequences.**

Corollary 1. *If the sequences r_i and s_i are finite (which means all elements are zero for some sufficiently large i), the continued fraction is terminating, therefore the generating function g is a rational polynomial of z and t.*

In Maple the computation of the generating function from the two lists r and s is implemented as follows:

```
# @param r list of r. Optionally with any number of trailing zeros.
# @param s list of s. Optionally with any number of trailing zeros.
# @param delfcol If true delete first column of resulting array
# @param delfrow If true delete first row of resulting array
# @return The generating function with unknowns  x and y.
DELTAgf := proc(r::list,s::list,x,y,delfcol::boolean,delfrow::boolean)
        local N,n,g;
        N := min(nops(r),nops(s)) ;
        g := 0 ;
        for n from N to 1 by -1 do
                (op(n,r)*x+op(n,s)*x*y)/(1-g) ;
                g := factor(%) ;
        end do:
        g := 1/(1-g) ;
        if delfcol then
                g := (g-subs(y=0,g))/y;
        end if;
        if delfrow then
                g := (g-subs(x=0,g))/x;
        end if;
        g := factor(g) ;
        printf("\nG.f.: (%a)/(%a). - ~~~~\n",factor(numer(g)),factor(denom(g))) ;
        g ;
end proc:
```

Remark 1. *This immediately establishes the generating functions of the cases A008288, A097806, A111049, A119865, A121314, A122542, A122935, A122950, A123110, A123149, A123585, A124645, A133607, A147703, A147721, A152815, A152842, A154388, A155161, A165253, A167374, A172250, A185331, A199479,* *A199856, A201730, A208324, A209599, A210239, A236376, and many more with two finite sequences r and s.*

3.3. **Finite Left or Right sequences.** If one of the two sequences r or s is finite and the continued fraction of the associated tail of the other has a known closed form, the generating function is obtained by plugging this closed form into the continued fraction. The best-known case is the 1-periodic Stieltjes continued fraction [5, (1.9.6)]:

$$1-\cfrac{\alpha x}{1-\cfrac{\alpha x}{1-\cfrac{\alpha x}{1-\cdots}}}=\frac{1+\sqrt{1-4\alpha x}}{2}. \tag{7}$$

Example 1. *Inserting $\alpha = 1$ gives*

$$(1+\sqrt{1-4x})/2 = 1-x-x^2-2x^3-5x^4-\cdots, \tag{8}$$

the Catalan numbers A000108.

Example 2. *The GF of A109450 is*

$$\cfrac{1}{1-\cfrac{zt}{1-\cfrac{z+zt}{(1+\sqrt{1-4zt})/2}}}=\frac{1-2(z+zt)+\sqrt{1-4zt}}{1-2z-3zt+(1-zt)\sqrt{1-4zt}}. \tag{9}$$

Example 3. *The GF of A106566 is*

$$\cfrac{1}{1-\cfrac{zt}{(1+\sqrt{1-4z})/2}}=\frac{1+\sqrt{1-4z}}{1-2zt+\sqrt{1-4z}}. \tag{10}$$

Further examples: A080247, A114193, A127543, A167685, A196182, A205813.

The general case of 2-periodic Stieltjes continued fractions is calculated by repeated inversion and subtraction of the continued fraction until a self-consistent quadratic equation emerges [12]:

$$1+\cfrac{\alpha x}{1+\cfrac{\beta x}{1+\cfrac{\alpha x}{1+\cfrac{\beta x}{1+\cdots}}}}=\frac{1+(\alpha-\beta)x+\sqrt{1+(\alpha-\beta)^2x^2+2(\alpha+\beta)x}}{2}. \tag{11}$$

(7) is recovered as a special case by setting $\alpha = \beta$. Binomial expansions of the square root have been discussed by Callan [3].

Example 4. *The 2-periodic example with alternating x and $2x$ in the numerators is*

$$1+\cfrac{x}{1+\cfrac{2x}{1+\cfrac{x}{1+\cdots}}}=1+x-2x^2+6x^3-22x^4+90x^5-\ldots=\frac{1-x+\sqrt{1+6x+x^2}}{2} \tag{12}$$

generating the Large Schröder numbers A006318 and providing a closed form of the GF of A011117, A080245, A104219, A108891, A132372, A133367, A172040, A172094.

Example 5. *The 2-periodic example with alternating x and $3x$ in the numerators is A047891:*

(13)

$$1+\cfrac{x}{1+\cfrac{3x}{1+\cfrac{x}{1+\cfrac{3x}{1+\cdots}}}} = 1+x-3x^2+12x^3-57x^4+300x^5-\ldots = \frac{1-2x-\sqrt{1+8x+4x^2}}{2}.$$

Example 6. *The generating function of A133366 is*

$$\frac{1}{1-\cfrac{3z+zt}{1-\cfrac{z}{1-\cfrac{3z}{1-\cfrac{z}{-}}}}} = \frac{1}{1-\cfrac{3z+zt}{(1+2z+\sqrt{1-8z+4z^2})/2}} \tag{14}$$

Example 7. *The 2-periodic example with alternating x and $4x$ in the numerators is A082298:*

(15)

$$1+\cfrac{x}{1+\cfrac{4x}{1+\cfrac{x}{1+\cfrac{4x}{1+\cdots}}}} = 1+x-4x^2+20x^3-116x^4+740^5-\ldots = \frac{1-3x+\sqrt{9x^2+10x+1}}{2}$$

Example 8. *The 2-periodic example with alternating x and $5x$ in the numerators is*

(16)

$$1+\cfrac{x}{1+\cfrac{5x}{1+\cfrac{x}{1+\cfrac{5x}{1+\cdots}}}} = 1+x-5x^2+30x^3-205x^4+1530x^5-\ldots = \frac{1-4x+\sqrt{1+16x^2+12x}}{2}$$

as described in A082301.

Example 9. *The 2-periodic example with alternating $2x$ and $3x$ in the numerators is*

(17)

$$1+\cfrac{2x}{1+\cfrac{3x}{1+\cfrac{2x}{1+\cfrac{3x}{1+\cdots}}}} = 1+2x-6x^2+30x^3-186x^4+1290x^5-\ldots = \frac{1-x+\sqrt{1+x^2+10x}}{2}$$

obtained by doubling the values of A103210.

Example 10. *The 2-periodic example with alternating $4x$ and $3x$ in the numerators is*

$$1+\cfrac{4x}{1+\cfrac{3x}{1+\cfrac{4x}{1+\cfrac{3x}{1+\cdots}}}} = 1+4x-12x^2+84x^3-732x^4+7140x^5-\dots \tag{18}$$

generated by multiplying the values of A131763 by 4.

Example 11. *The 2-periodic example with alternating $-2x$ and x in the numerators is*

$$1-2x+2x^2+2x^3-2x^4-10x^5-\dots = \frac{1-3x+\sqrt{1+9x^2-2x}}{2} \tag{19}$$

of A152681.

Example 12. *The 2-periodic example with alternating $-2x$ and $2x$ in the numerators is*

$$1-2x+4x^2-16x^4+128x^6-1280x^8+\dots = \frac{1-4x+\sqrt{1+16x^2}}{2} \tag{20}$$

obtained by multiplying the entries of A151403 by 4 and alternating signs.

The 3-periodic Stieltjes continued fractions are

$$1+\cfrac{\alpha x}{1+\cfrac{\beta x}{1+\cfrac{\gamma x}{1+\cfrac{\alpha x}{1+\cfrac{\beta x}{1+\cdots}}}}} = \frac{1+(\alpha+\beta-\gamma)x+\sqrt{[1+(\alpha+\beta-\gamma)x]^2+4\gamma x(1+\alpha x)(1+\beta x)}}{2(1+\beta x)}. \tag{21}$$

Equation (7) is recovered as a special case by letting $\beta=\gamma=\alpha$.

Example 13. *Setting $\alpha=1$, $\beta=2$, $\gamma=1$ gives*

$$1+x-2x^2+6x^3-20x^4+72x^5-276x^6+112x^7-\dots = \frac{1+2x+\sqrt{1+8x+16x^2+8x^3}}{2(1+2x)}, \tag{22}$$

a variant of A059279.

Example 14. *Setting $\alpha=1$, $\beta=\gamma=2$ gives*

$$1+x-2x^2+8x^3-36x^4+172x^5-860x^6+4460x^7-\dots = \frac{1+x+\sqrt{1+10x+25x^2+16x^3}}{2(1+2x)}, \tag{23}$$

a variant of A186338.

Example 15. *Setting $\alpha=-1$, $\beta=2$, $\gamma=1$ gives*

$$1-x+2x^2-6x^3+16x^4-40x^5+92x^6-\dots = \frac{1+\sqrt{1+4x+4x^2-8x^3}}{2(1+2x)}, \tag{24}$$

a variant of A174016.

Example 16. *Setting $\alpha = \beta = 1$ and $\gamma = -1$ gives*
(25)
$$1+x-x^2+x^4-2x^5+2x^6-5x^8+12x^9-16x^{10}\cdots = \frac{1+3x+\sqrt{1+2x+x^2-4x^3}}{2(1+x)},$$

a variant of A168505.

Example 17. *Setting $\alpha = 1$, $\beta = \gamma = -1$ gives*
(26)
$$1+x+x^2+2x^3+3x^4+4x^5+4x^6+2x^7-3x^8+10x^9-14x^{10}\cdots = \frac{1+x+\sqrt{1-2x+x^2+4x^3}}{2(1-x)},$$

described in A174015 and used in A174014.

Example 18. *Inserting $r_i = +1, +2, +1, +3, +1, +4, \ldots$ gives the g.f. described in A090365.*

3.4. **Doubly Periodic Left and Right sequences.** If the r_i and also the s_i have a common period length, the generating function also falls into the category of multi-periodic Stieltjes continued fractions.

Example 19. *In sequence A157491 r_i is 1-periodic $0, -1, -1, -1, \ldots$ and s_i is 1-periodic $1, 1, 1, \ldots$. The GF is with $\alpha = zt - z$ and $x = 1$ in (7)*

$$\frac{1+\sqrt{1+4z-4zt}}{1-2zt+\sqrt{1+4z-4zt}}. \tag{27}$$

Example 20. *In sequence A174014 r_i is 3-periodic $1, 1, -1, \ldots$ and s_i is 3-periodic $1, 0, 0, \ldots$. The GF is with (21) and $\alpha = -1-t$, $\beta = -1$, $\gamma = 1$, $x = z$*

$$\frac{2(1-z)}{1-(3+t)z+\sqrt{1-2z(1+t)+z^2(1+t)^2+4z^3(1+t)}}. \tag{28}$$

Example 21. *In sequence A131198 r_i is 2-periodic $1, 0, \ldots$ and s_i is 2-periodic $0, 1, \ldots$. The GF is with (11) and $\alpha = -1$, $\beta = -t$, $x = z$*

$$\frac{2}{1+(t-1)z+\sqrt{1+z^2(1-t)^2-2z(1+t)}}. \tag{29}$$

The technique yields closed form GF's for A060693, A085880, A090981, A091977, A094385, A104684, A126216, A114608, A114656, A114687, A123254, A127529, A133336, A175136, A198379. The 4-periodic analogue of (21) is

$$1+\cfrac{\alpha x}{1+\cfrac{\beta x}{1+\cfrac{\gamma x}{1+\cfrac{\delta x}{1+\cfrac{\alpha x}{1+\cdots}}}}} = \frac{\beta x+(1+\alpha x)(1+\gamma x)-\delta x(1+\beta x)}{2(1+\beta x+\gamma x)} \tag{30}$$
$$+\frac{\sqrt{[\beta x+(1+\alpha x)(1+\gamma x)-\delta x(1+\beta x)]^2+4\delta x(1+\alpha x+\beta x)(1+\beta x+\gamma x)}}{2(1+\beta x+\gamma x)}.$$

Setting $\gamma = \alpha$ and $\delta = \beta$ recovers (11) as a special case.

Example 22. *With $\alpha = \beta = -zt$, $\gamma = \delta = -z$ and $x = 1$ the closed form for A168511 is*

$$\frac{2(1-zt-z)}{1-2zt-\sqrt{(1-2z)(1-2zt)(1-2z-2zt)}}. \tag{31}$$

Example 23. *With $\alpha = -zt$, $\beta = -z$, $\gamma = zt$, $\delta = z$ and $x = 1$ the GF of A172101 is*

$$\frac{2(1-z+zt)}{(1-z)^2-z^2t^2+\sqrt{(1+z+zt)(1-z+zt)(1-z-zt)(1+z-zt)}}. \tag{32}$$

By subtracting 1 and dividing through zt one obtains the GF of A088855.

4. Inverse Operator

The inverse operator takes a number triangle $T_{n,k}$ and calculates the two sequences r_i and s_i. We note that the first N elements of r plus the first N elements of s fix the lower left triangle $T_{n,k}$ up to row N with a total of $(N+1)(N+2)/2$ elements. This grows roughly proportional to the N^2, so it is obvious that not every number triangle can be generated by the Δ-operator. So the inverse operation to map the elements $T_{n,k}$, $0 \le n \le N$ onto the (r_i, s_i), $i \le N$ may not exist.

The algorithm to compute the r_i given the T can be based on the fact that the left column $T_{n,0}$ is generated by $g(z,0)$, setting $t = 0$ in (5) and Theorem (4) [1, §5]. This means if the sequence of the elements $T_{0,i}$ is known and if $T_{0,0} = 1$ then the r_i are known by recursive lookup of [8]

$$\sum_{n\ge 0} T_{n,0} z^n = \cfrac{1}{1-\cfrac{r_0 z}{1-\cfrac{r_1 z}{1-\cfrac{r_2 z}{1-\cdots}}}}. \tag{33}$$

Again, this inverse may not exist for some sequence $T_{n,0}$ because some of the intermediate r_i may turn out to be zero [8]. In Maple this may be implemented by repeated inversion of Taylor series (as in Example III of [11]):

```
# Given L=[1,a1,a2,..] representing 1+a1*x+a2*x^2+..
# compute the Stieltjes fractions 1+s1x/(1+s2*x/(1+s3*x/...)
# @param L The list of [1,a1,a2,a3,...]
# @return The list of [s1,s2,s3,...]
# @since 2015-08-15
# @author R. J. Mathar
sfrac := proc(L::list)
        local slen,S,Lred,x ;
        slen := nops(L) ;
        if op(1,L) <> 1 then
                error "first element", op(1,L)," not unity"
        end if;
        S := [op(2,L)] ;
        if slen > 2 and op(2,L) <> 0 then
                # 1+a1*x+a2*x^2+...=1+a1*x(1+b1*x+b2*x^2+...)
                # At this point a premature division through zero indicates
                # that the inverse (the Stieltjes cf) does not exist.
                Lred := [seq(op(i,L)/op(2,L),i=2..slen)] ;
                # rewrite 1+b1*x+b2*x^2+.. = 1+b1*x/(1+c1*x+c2*x^2+..)
```

```
                gfun[listtoseries](Lred,x) ;
                taylor(1/%,x=0,slen) ;
                gfun[seriestolist](%) ;
                procname(%) ;
                S := [op(S),op(%)] ;
        end if;
        return S;
end proc:

# Given L=[1,a1,a2,a3,...] representing 1+a1*x+a2*x^2+..
# compute the Stieltjes fractions 1/(1-r1*x/(1-r2*x/(1-r3*x/...).
# This does essentially the same as sfrac() but flips the signs of the numerators
# of the continued fraction and applies one more level of 1/(1-..).
# @param L The list of [1,a1,a2,a3,...]
# @return The list of [r1,r2,r3,...]
# @since 2015-08-15
# @author R. J. Mathar
sfracDelta := proc(L::list)
        local slen,S,x ;
        slen := nops(L) ;
        if op(1,L) <> 1 then
                error "first element", op(1,L)," not unity"
        end if;
        gfun[listtoseries](L,x) ;
        taylor(1/%,x=0,slen) ;
        gfun[seriestolist](%) ;
        S := sfrac(%) ;
        [seq(-op(i,S),i=1..nops(S))] ;
        return %;
end proc:
```

The numerators r_i may also be expressed as ratios of determinants built from the series coefficients $T_{n,0}$ [7, 6, 8]. An alternative is to construct a diagonal of the Padé table as an intermediate step and to proceed with an auxiliary Routh array [9, 10, 4, 2].

The elements of s_i are obtained from (6) by focussing on the coefficients of equal powers in z and t and repeating the algorithm of the r_i, but this time the input taken from the diagonal of the triangle [1, §5]:

$$\sum_{n\ge 0}(zt)^n T_{n,n} = \cfrac{1}{1-\cfrac{s_0 zt}{1-\cfrac{s_1 zt}{1-\cfrac{s_2 zt}{1-\cdots}}}}. \tag{34}$$

5. Riordan Arrays

The generating function of the k-th column of an array of the element $(g(z), f(z))$ of the Riordan group is [1]

$$g(z)f(z)^k = \sum_n T_{n,k} z^n. \tag{35}$$

So the generating function of the triangle is the geometric series

$$\sum_k t^k \sum_n T_{n,k} z^n = \frac{g(z)}{1 - tf(z)}. \tag{36}$$

If $g(z) = 1$ and the order of $f(z) = s_0 z + \cdots$ is 1, this generating function matches (6) with $r = 0, \ldots$ and $s = s_0, 0, 0, 0, \ldots$, which means the triangle is representable by the Delta-operator [1, §5.3]. Examples of this association are A090238, A106566, A108747, A122538, A122542, A147721, A155161, A172040, A204533, A205813, A206022, A206294, A206306, A207327, A220399, A221179, and many others.

References

1. Paul Barry, *A study of integer sequences, riordan arrays, pascal-like arrays and hankel transforms*, Ph.D. thesis, University College, Cork, 2009.
2. M. J. Bosley, H. W. Kropholler, and F. P. Lees, *On the relation between the continued fraction expansion and moments matching methods of model reduction*, Intl. J. Control **18** (1973), no. 3, 461–474.
3. David Callan, *On generating functions involving the square root of a quadratic polynomial*, J. Integer Seq. **10** (2007), # 07.5.2. MR 2304410 (2007m:05011)
4. Chi-Fan Chen and Leang-San Shieh, *Continued fraction inversion by routh's algorithm*, IEEE Trans. Circ. Theory **16** (1969), no. 2, 197–202.
5. Annie A. M. Cuyt, Vigdis Brevik Petersen, Brigitte Verdonk, Haakon Waadeland, and William B. Jones, *Handbook of continued fractions for special functions*, Springer, New York, 2008. MR 2410517
6. William B. Gragg, *Matrix interpretations and applications of the continued fraction algorithm*, Rocky Mount. J. Math. **4** (1974), no. 2, 213.
7. J. B. H. Heilermann, *Ueber die verwandlung der reihen in kettenbrüche*, J. Reine Angew. Mathem. **33** (1846), 174–188.
8. Walter Leighton and W. T. Scott, *A general continued fraction expansion*, Bull. Am. Math. Soc. **45** (1939), no. 8, 596–605.
9. I. M. Longman, *Computation of the Padé table*, Intl. J. Comput. Math. **3** (1972), no. 1–4, 53–64.
10. Arne Magnus, *Continued fractions associated with the padé table*, Math. Zeitschr. **78** (1962), 361–374.
11. L. S. Shieh, W. P. Schneider, and D. R. Williams, *A chain of factored matrices for routh array and continued fraction inversion*, Int. J. Control **13** (1971), no. 4, 691–703.
12. H. Siebeck, *Ueber periodische kettenbrüche*, J. Reine Angew. Mathem. **33** (1846), 68–70.
13. Neil J. A. Sloane, *The On-Line Encyclopedia Of Integer Sequences*, Notices Am. Math. Soc. **50** (2003), no. 8, 912–915, http://oeis.org/. MR 1992789 (2004f:11151)

E-mail address: `mathar@mpia.de`
URL: `http://www.mpia.de/~mathar`

Max-Planck Institute of Astronomy, Königstuhl 17, 69117 Heidelberg, Germany